Computation without Representation

Thanks for inviting me to the SCGP!

ωεβ:=http://drorbn.net/sb18/

Follows Rozansky [Ro1, Ro2, Ro3] and Overbay [Ov], joint with van der Veen. More at [BV] and at ωεβ/talks.

Abstract. A major part of "quantum topology" is the definition and computation of various knot invariants by carrying out computations in quantum groups. Traditionally these computations are carried out "in a representation", but this is very slow: one has to use tensor powers of these representations, and the dimensions of powers grow exponentially fast. I will describe a direct-participation method for carrying out these computations without having to choose a representation and explain why in many ways the results are better and faster. The two key points we use are a technique for composing infinite-order "perturbed Gaussian" differential operators, and the little-known fact that every semi-simple Lie algebra can be approximated by solvable Lie algebras, where computations are easier.

A Knot Theory Portfolio.

- Has operations $\sqcup, m, \Delta, S$.
- All tangles are generated by $R^{\pm 1}$ and $C^{\pm}$ (so "easy" to produce invariants).
- Makes some properties ("genus", "ribbon") be "definable".

(more to say, but not now).



A "Quantum Group" Portfolio consists of an algebra U along with maps (and some axioms...)



PBW Bases. The U's we care about always have "Poincaré-Birkhoff-Witt" bases; there is some finite set B of "generators" and isomorphisms $\mathbb{O}_{y,x,\ldots}\colon \hat{\mathcal{S}}(B) \to U$ defined by "ordering monomials" to some fixed $y, x, \ldots$ order. The quantum group portfolio now becomes a "symmetric algebra" portfolio, or a "power series" portfolio.

The (Semi-)Category $\mathcal{DO}$. Hence we care about the monoidal (semi-)category $\mathcal{DO}$ whose objects are finite sets B and whose morphisms are $\mathrm{mor}_{\mathcal{DO}}(B, B') := \mathrm{Hom}_{\mathbb{Q}}(\mathcal{S}(B) \to \mathcal{S}(B')) = \mathcal{S}(B^*, B')$ (by convention, $x^* = \xi$, $y^* = \eta$, etc.).

The Composition Law. If

$$\mathcal{S}(B_0) \xrightarrow[{}^t f \in \mathbb{Q}[\![\zeta_{0i}, z_{1j}]\!]]{f} \mathcal{S}(B_1) \xrightarrow[{}^t g \in \mathbb{Q}[\![\zeta_{1j}, z_{2k}]\!]]{g} \mathcal{S}(B_2)$$

then ${}^t(f/\!\!/g) = {}^t(g \circ f) = \left(g|_{\zeta_{1j} \to \partial_{z_{1j}}} f \right)_{z_{1j}=0} = \left(f|_{z_{1j} \to \partial_{\zeta_{1j}}} g \right)_{\zeta_{1j}=0}$.

Note. For $f \in \mathcal{S}(z)$ and $g \in \mathcal{S}(\zeta)$, $\langle f, g \rangle = f(\partial_\zeta) g|_{\zeta=0} = g(\partial_z) f|_{z=0}$.

Proposition. If $F\colon \mathcal{S}(B) \to \mathcal{S}(B')$ is linear and "continuous", then ${}^t F = \exp\left(\sum_{z_i \in B} \zeta_i z_i\right) /\!\!/ F$.

Examples.

1. The 1-variable identity map $I\colon \mathcal{S}(z) \to \mathcal{S}(z)$ is given by ${}^t I_1 = \mathbb{e}^{z\zeta}$ and the n-variable one by ${}^t I_n = \mathbb{e}^{z_1\zeta_1 + \cdots + z_n\zeta_n}$.
2. The "$z_i \to z_j$ variable rename map $\sigma^i_j\colon \mathcal{S}(z_i) \to \mathcal{S}(z_j)$" becomes ${}^t\sigma^i_j = \mathbb{e}^{z_j\zeta_i}$, and it's easy to rename several variables simultaneously.
3. The "archetypal multiplication map $m^{ij}_k\colon \mathcal{S}(z_i, z_j) \to \mathcal{S}(z_k)$" has ${}^t m = \mathbb{e}^{z_k(\zeta_i + \zeta_j)}$.
4. The "archetypal coproduct $\Delta^i_{jk}\colon \mathcal{S}(z_i) \to \mathcal{S}(z_j, z_k)$", given by $z_i \to z_j + z_k$ or $\Delta z = z \otimes 1 + 1 \otimes z$, has ${}^t\Delta = \mathbb{e}^{(z_j + z_k)\zeta_i}$.
5. R-matrices tend to have terms of the form $\mathbb{e}_q^{\hbar y_1 x_2} \in \mathcal{U}_q \otimes \mathcal{U}_q$. The "baby R-matrix" is ${}^t R = \mathbb{e}^{\hbar yx} \in \mathcal{S}(y, x)$.
6. The "Weyl form of the canonical commutation relations" states that if $[y, x] = t$ is a scalar, then $\mathbb{e}^{\xi x}\mathbb{e}^{\eta y} = \mathbb{e}^{\eta y}\mathbb{e}^{\xi x}\mathbb{e}^{-\eta\xi t}$. Thus with



we have ${}^t SW_{xy} = \mathbb{e}^{\eta y + \xi x - \eta\xi t}$.

The Zipping Issue. (between unbound and bound lies half-zipped).



Zipping. If $P(\zeta^j, z_i)$ is a polynomial, or whenever otherwise convergent, set

$$\left\langle P(\zeta^j, z_i) \right\rangle_{(\zeta^j)} = P\left(\partial_{z_j}, z_i\right)\Big|_{z_i=0}.$$

(E.g., if $P = \sum a_{nm}\zeta^n z^m$ then $\langle P \rangle_\zeta = \sum n! a_{nn}$).

Implementation. ωεβ/Zip

```
z* = ζ; ζ* = z; Zip{}[P_] := P;
Zip{ζ_,ζs___}[P_] :=
  (Expand[P // Zip{ζs}] /. f_. ζ^d_. :→ ∂{ζ*,d}f) /. ζ* → 0
{Zip{ζ}[ζ² e^(δ z²)], Zip{ζ}[ζ⁴ e^(δ z²)]}
```

{2 δ, 12 δ²}



"God created the knots, all else in topology is the work of mortals."

Leopold Kronecker (modified)

www.katlas.org

The Zipping / Contraction Theorem. If P has a finite ζ-degree and the y's and the q's are "small" then

$$\left\langle P(z_i,\zeta^j)\mathbb{e}^{\eta^i z_i + y_j\zeta^j}\right\rangle_{(\zeta^j)} = \left\langle P(z_i+y_i,\zeta^j)\mathbb{e}^{\eta^i(z_i+y_i)}\right\rangle_{(\zeta^j)},$$

(proof: replace $y_j \to \hbar y_j$ and test at $\hbar = 0$ and at $\partial_\hbar$), and

$$\left\langle P(z_i,\zeta^j)\mathbb{e}^{c+\eta^i z_i+y_j\zeta^j+q^i_j z_i\zeta^j}\right\rangle_{(\zeta^j)} = \det(\tilde q)\left\langle P(z_i,\zeta^j)\mathbb{e}^{c+\eta^i z_i}\Big|_{z_i\to\tilde q^k_i(z_k+y_k)}\right\rangle_{(\zeta^j)}$$

where $\tilde q$ is the inverse matrix of $1-q$: $(\delta^i_j - q^i_j)\tilde q^j_k = \delta^i_k$ (proof: replace $q^i_j \to \hbar q^i_j$ and test at $\hbar = 0$ and at $\partial_\hbar$).

Exponential Reservoirs.. The true Hilbert hotel is exp! Remove one x from an "exponential reservoir" of x's and you are left with the same exponential reservoir:

$$\mathbb{e}^x = \left[\ldots + \frac{xxxxx}{120} + \ldots\right] \xrightarrow{\partial_x} \left[\ldots + \frac{x\!\!\!/xxxx}{120} + \ldots\right] = (\mathbb{e}^x)' = \mathbb{e}^x,$$

and if you let each element choose left or right, you get twice the same reservoir:

$$\mathbb{e}^x \xrightarrow{x\to x_l+x_r} \mathbb{e}^{x_l+x_r} = \mathbb{e}^{x_l}\mathbb{e}^{x_r}.$$



A Graphical Proof. Glue top to bottom on the right, in all possible ways. Several scenarios occur:

1. Start at A, go through the q-machine $k \geq 0$ times, stop at B. Get $\left\langle P\left(\sum_{k\geq 0} q^k z,\zeta\right)\right\rangle = \langle P(\tilde q z,\zeta)\rangle$.
2. Loop through the q-machine and swallow your own tail. Get $\exp\left(\sum q^k/k\right) = \exp(-\log(1-q)) = \tilde q$.
3. …

By the reservoir splitting principle, these scenarios contribute multiplicatively. □

Implementation. ωεβ/Zip

```
𝔼 /: Zip_{ξs_List}@𝔼[Q_, P_] :=      (* 𝔼[Q,P] means e^Q P *)
  Module[{ξ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ξ*, {ξ, ξs}];
   c = Q /. Alternatives @@ (ξs ⋃ zs) → 0;
   ys = Table[∂_ξ (Q /. Alternatives @@ zs → 0), {ξ, ξs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ξs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ξ*} − ∂_{z,ξ}Q, {ξ, ξs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q1 = c + ηs.zs /. zrule; Q2 = Q1 /. Alternatives @@ zs → 0;
   Simplify /@ 𝔼[Q2, Det[qt] e^{-Q2} Zip_{ξs}[e^{Q1} (P /. zrule)]] ];
```

Fuller program & testing suite: ωεβ/mm, ωεβ/port.

The Real Thing. In the algebra QU_ϵ (explained later), over $\mathbb{Q}[[\hbar]]$ using the $yaxt$ order, $T = \mathbb{e}^{\hbar t}$, $\bar T = T^{-1}$, $\mathcal{A} = \mathbb{e}^{\alpha}$, and $\bar{\mathcal{A}} = \mathcal{A}^{-1}$, we have

$${}^t R_{ij} = \mathbb{e}^{\hbar(y_i x_j - t_i a_j)}\left(1 + \epsilon\hbar\left(a_i a_j - \hbar^2 y_i^2 x_j^2/4\right) + O(\epsilon^2)\right)$$

in $\mathcal{S}(B_i, B_j)$, and in $\mathcal{S}(B_1^*, B_2^*, B)$ we have

$${}^t m = \mathbb{e}^{(\alpha_1+\alpha_2)a + \eta_2\xi_1(1-T)/\hbar + (\xi_1\bar{\mathcal{A}}_2+\xi_2)x + (\eta_1+\eta_2\bar{\mathcal{A}}_1)y}\left(1 + \epsilon\lambda + O(\epsilon^2)\right),$$

where $\lambda = 2a\eta_2\xi_1 T + \eta_2^2\xi_1^2\left(3T^2 - 4T + 1\right)/4\hbar - \eta_2\xi_1^2(3T-1)x\bar{\mathcal{A}}_2/2 - \eta_2^2\xi_1(3T-1)y\bar{\mathcal{A}}_1/2 + \eta_2\xi_1 xy\hbar\bar{\mathcal{A}}_1\bar{\mathcal{A}}_2$.

Finally,

$${}^t\Delta = \mathbb{e}^{\tau(t_1+t_1)+\eta(y_1+T_1 y_2)+\alpha(a_1+a_2)+\xi(x_1+x_2)}\left(1 + O(\epsilon)\right) \in \mathcal{S}(B^*, B_1, B_2),$$

and ${}^t S = \mathbb{e}^{-\tau t - \alpha a - \eta\xi(1-\bar T)\mathcal{A}/\hbar - \bar T\eta y\mathcal{A} - \xi x\mathcal{A}}\left(1 + O(\epsilon)\right) \in \mathcal{S}(B^*, B)$.

Real Zipping is a minor mess, and is done in two phases:

	τa-phase		ξy-phase	
ζ-like variables	τ	a	ξ	y
z-like variables	t	α	x	η

Already at $\epsilon = 0$ we get the best known formulas for the Alexander polynomial!

Generic Docility. A "docile perturbed Gaussian" in the variables $(z_i)_{i\in S}$ over the ring R is an expression of the form

$$\mathbb{e}^{q^{ij}z_i z_j}P = \mathbb{e}^{q^{ij}z_i z_j}\left(\sum_{k\geq 0}\epsilon^k P_k\right),$$

where all coefficients are in R and where P is a "docile series": $\deg P_k \leq 4k$.

Our Docility. In the case of QU_ϵ, all invariants and operations are of the form $\mathbb{e}^{L+Q}P$, where

- L is a quadratic of the form $\sum l_{z\zeta}z\zeta$, where z runs over $\{t_i, \alpha_i\}_{i\in S}$ and ζ over $\{\tau_i, a_i\}_{i\in S}$, with integer coefficients $l_{z\zeta}$.
- Q is a quadratic for the form $\sum q_{z\zeta}z\zeta$, where z runs over $\{x_i, \eta_i\}_{i\in S}$ and ζ over $\{\xi_i, y_i\}_{i\in S}$, with coefficients $q_{z\zeta}$ in the ring R_S of rational functions in $\{T_i, \mathcal{A}_i\}_{i\in S}$.
- P is a docile power series in $\{y_i, a_i, x_i, \eta_i, \xi_i\}_{i\in S}$ with coefficients in R_S, and where $\deg(y_i, a_i, x_i, \eta_i, \xi_i) = (1, 2, 1, 1, 1)$.

Docililty Matters! The rank of the space of docile series to ϵ^k is polynomial in the number of variables $|S|$. !!!!!

- At $\epsilon^2 = 0$ we get the Rozansky-Overbay [Ro1, Ro2, Ro3, Ov] invariant, which is stronger than HOMFLY-PT polynomial and Khovanov homology taken together!
- In general, get "higher diagonals in the Melvin-Morton-Rozansky expansion of the coloured Jones polynomial" [MM, BNG], but why spoil something good?

Solvable Approximation. In gl_n, half is enough! Indeed $gl_n \oplus \mathfrak{a}_n = \mathcal{D}(◹, b, \delta)$:



$b(◹) = b$: ◹⊗◹ → ◹
$b(◺) \rightsquigarrow \delta$: ◹ → ◹⊗◹

Now define $gl_n^\epsilon := \mathcal{D}(◹, b, \epsilon\delta)$. Schematically, this is $[◹, ◹] = ◹$, $[◺, ◺] = \epsilon◺$, and $[◹, ◺] = ◺ + \epsilon◹$. The same process works for all semi-simple Lie algebras, and at $\epsilon^{k+1} = 0$ always yields a solvable Lie algebra.

CU and *QU*. Starting from sl_2, get $CU_\epsilon = \langle y, a, x, t\rangle/([t, -] = 0, [a, y] = -y, [a, x] = x, [x, y] = 2\epsilon a - t)$. Quantize using standard tools (I'm sorry) and get $QU_\epsilon = \langle y, a, x, t\rangle/([t, -] = 0, [a, y] = -y, [a, x] = x, xy - \mathbb{e}^{\hbar\epsilon}yx = (1 - T\mathbb{e}^{-2\hbar\epsilon a})/\hbar)$.

References.

[BNG] D. Bar-Natan and S. Garoufalidis, *On the Melvin-Morton-Rozansky conjecture,* Invent. Math. **125** (1996) 103–133.

[BV] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* arXiv:1708.04853.

[MM] P. M. Melvin and H. R. Morton, *The coloured Jones function,* Commun. Math. Phys. **169** (1995) 501–520.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* University of North Carolina PhD thesis, ωεβ/Ov.

[Ro1] L. Rozansky, *A contribution of the trivial flat connection to the Jones polynomial and Witten's invariant of 3d manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

The Algebras H and H^*. Let $q = e^{\epsilon}$ and set $H = \langle a, x\rangle/([a,x] = x)$ with

$$A = e^{-\epsilon a}, \quad xA = qAx, \quad S_H(a, A, x) = (-a, A^{-1}, -A^{-1}x),$$
$$\Delta_H(a, A, x) = (a_1 + a_2, A_1A_2, x_1 + A_1x_2)$$

and dual $H^* = \langle b, y\rangle/([b, y] = -\epsilon y)$ with

$$B = e^{-b}, \quad By = qyB, \quad S_{H^*}(b, B, y) = (-b, B^{-1}, -yB^{-1}),$$
$$\Delta_{H^*}(b, B, y) = (b_1 + b_2, B_1B_2, y_1B_2 + y_2).$$

Pairing by $(a, x)^* = (b, y)$ ($\Rightarrow \langle B, A\rangle = q$) making $\langle y^l b^i, a^j x^k\rangle = \delta_{ij}\delta_{kl}\, j![k]_q!$ so $R = \sum \frac{y^k b^j \otimes a^j x^k}{j![k]_q!}$.

The Algebra QU. By the Drinfel'd double procedure, $QU = H^{*cop} \otimes H$ with $(\phi f)(\psi g) = \langle \psi_1 S^{-1} f_3\rangle\langle\psi_3, f_1\rangle(\phi\psi_2)(f_2 g)$ and

$$S(y, b, a, x) = (-B^{-1}y, -b, -a, -A^{-1}x),$$
$$\Delta(y, b, a, x) = (y_1 + y_2B_1, b_1 + b_2, a_1 + a_2, x_1 + A_1x_2).$$

The 2D Lie Algebra. Clever people know* that if $[a, x] = \gamma x$ then $e^{\xi x}e^{\alpha a} = e^{\alpha a}e^{e^{-\gamma\alpha}\xi x}$. Ergo with



we have ${}^t SW_{ax} = e^{\alpha a + e^{-\gamma\alpha}\xi x}$.

* Indeed $xa = (a - \gamma)x$ thus $xa^n = (a-\gamma)^n x$ thus $xe^{\alpha a} = e^{\alpha(a-\gamma)}x = e^{-\gamma\alpha}e^{\alpha a}x$ thus $x^n e^{\alpha a} = e^{\alpha a}(e^{-\gamma\alpha})^n x^n$ thus $e^{\xi x}e^{\alpha a} = e^{\alpha a}e^{e^{-\gamma\alpha}\xi x}$.

A Full Implementation at $\epsilon^2 = 0$.

```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@
    Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
```

```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] :=
  𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z};
(u_i_)* := (u*)_i;
```

```
expand[sd_SeriesData] := MapAt[expand, sd, 3];
expand[ℰ_] := Expand[ℰ];
Zip_{}[P_] := P;
Zip_{ζ_,ζs___}[P_] :=
 (expand[P // Zip_{ζs}] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] :=
  Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zip_ζs[e^Q1 (P /. zrule)]]];
QZip_{ζs_List} := QZip_{ζs,CF};
```

```
U2l = {B_i_^p_. → e^(-p b_i), B^p_. → e^(-p b), T_i_^p_. → e^(p t_i), T^p_. → e^(p t),
   𝒜_i_^p_. → e^(p α_i), 𝒜^p_. → e^(p α)};
l2U = {e^(c_. b_i_ + d_.) :→ B_i^-c e^d, e^(c_. b + d_.) :→ B^-c e^d,
   e^(c_. t_i_ + d_.) :→ T_i^c e^d, e^(c_. t + d_.) :→ T^c e^d,
   e^(c_. α_i_ + d_.) :→ 𝒜_i^c e^d, e^(c_. α + d_.) :→ 𝒜^c e^d,
   e^ℰ_ :→ e^Expand@ℰ};
```

```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] :=
  Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
   simp /@
    𝔼[L2, Q2, Det[lt] e^(-L2-Q2)
      Zip_ζs[e^(L1+Q1) (P /. U2l /. zrule)]] //. l2U];
LZip_{ζs_List} := LZip_{ζs,CF};
```

```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
   Times[
     L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i},
       {i, {is}}],
     R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Flatten@Table[{β_{n@i}, τ_{n@i}, a_{n@i}}, {i, {is}}]} //
   QZip_{Flatten@Table[{ξ_{n@i}, y_{n@i}}, {i, {is}}]}];
B_{l_List}[L_, R_] := Bind_l[L, R];
B_{is___}[L_, R_] := Bind_{is}[L, R];
```

```
am_{i_,j_->k_} := 𝔼[(α_i + α_j) a_k, (e^(-α_j) ξ_i + ξ_j) x_k, 1 + O[ϵ]^2]
aΔ_{i_->j_,k_} := 𝔼[α_i (a_j + a_k), ξ_i (x_j + x_k),
   1 + ϵ ξ_i x_k (-a_j + ξ_i x_j / 2) + O[ϵ]^2]
aS_{i_} := 𝔼[-α_i a_i, - e^(α_i) ξ_i x_i,
   1 - ϵ e^(α_i) ξ_i x_i (a_i + e^(α_i) ξ_i x_i / 2) + O[ϵ]^2]
aSi_{i_} := 𝔼[-α_i a_i, -e^(α_i) ξ_i x_i,
   1 - ϵ e^(α_i) ξ_i x_i (a_i - 1 + e^(α_i) ξ_i x_i / 2) + O[ϵ]^2]
```

```
bm_{i_,j_→k_} := 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, 1 - ϵ η_j y_k β_i + O[ϵ]^2]
bΔ_{i_->j_,k_} := 𝔼[β_i (b_j + b_k), η_i (e^(-b_k) y_j + y_k),
   1 + ϵ η_i^2 y_j y_k e^(-b_k) / 2 + O[ϵ]^2]
bS_{i_} := 𝔼[-β_i b_i, -e^(b_i) η_i y_i,
   1 - ϵ e^(b_i) η_i y_i (β_i + e^(b_i) η_i y_i / 2) + O[ϵ]^2]
bSi_{i_} := 𝔼[-β_i b_i, -e^(b_i) η_i y_i,
   1 - ϵ e^(b_i) η_i y_i (β_i - 1 + e^(b_i) η_i y_i / 2) + O[ϵ]^2]
```

```
tP_{i_,j_} := 𝔼[β_i α_j, η_i ξ_j, 1 + ϵ η_i^2 ξ_j^2 / 4]
```

```
Block[{i, j, k},
 dm_{i_,j_→k_} =
  (𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (aΔ_{i→h1,h2} ~B_{h2}~ aΔ_{h2→h2,h3})
     (bΔ_{j→t1,t2} ~B_{t2}~ bΔ_{t2→t2,t3})) ~B_{h3}~ aSi_{h3} ~B_{t1,h3}~
   (tP_{t1,h3}) ~B_{t3,h1}~ (tP_{t3,h1}) ~B_{h2,j,i,t2}~ (am_{h2,j→k} bm_{i,t2→k})]
```

$$\mathbb{E}\Big[a_k\alpha_i + a_k\alpha_j + b_k\beta_i + b_k\beta_j,\ \frac{1}{\mathcal{A}_i\mathcal{A}_j}\big(y_k\mathcal{A}_i\mathcal{A}_j\eta_i + y_k\mathcal{A}_j\eta_j + x_k\mathcal{A}_i\xi_i + \mathcal{A}_i\mathcal{A}_j\eta_j\xi_i - B_k\mathcal{A}_i\mathcal{A}_j\eta_j\xi_i + x_k\mathcal{A}_i\mathcal{A}_j\xi_j\big),\ 1 + \frac{1}{4\mathcal{A}_i\mathcal{A}_j}\big(-4y_k\mathcal{A}_j\beta_i\eta_j - 4x_k\mathcal{A}_i\beta_j\xi_i + 4x_ky_k\eta_j\xi_i + 4a_kB_k\mathcal{A}_i\mathcal{A}_j\eta_j\xi_i + 2y_k\mathcal{A}_j\eta_j^2\xi_i - 6B_ky_k\mathcal{A}_j\eta_j^2\xi_i + 2x_k\mathcal{A}_i\eta_j\xi_i^2 - 6B_kx_k\mathcal{A}_i\eta_j\xi_i^2 + \mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2 - 4B_k\mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2 + 3B_k^2\mathcal{A}_i\mathcal{A}_j\eta_j^2\xi_i^2\big)\epsilon + O[\epsilon]^2\Big]$$

```
Block[{i},
 dS_{i_} = 𝔼[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] ~B_{1,2}~ (bSi_1 aS_2) ~
   B_{1,2}~dm_{2,1→i}]
```

$$\mathbb{E}\Big[-a_i\alpha_i - b_i\beta_i,\ \frac{-y_i\mathcal{A}_i\eta_i - B_ix_i\mathcal{A}_i\xi_i + \mathcal{A}_i\eta_i\xi_i - B_i\mathcal{A}_i\eta_i\xi_i}{B_i},\ 1 + \frac{1}{4B_i^2}\big(4B_iy_i\mathcal{A}_i\eta_i - 4B_iy_i\mathcal{A}_i\beta_i\eta_i - 2y_i^2\mathcal{A}_i^2\eta_i^2 - 4a_iB_i^2x_i\mathcal{A}_i\xi_i - 4B_i^2x_i\mathcal{A}_i\beta_i\xi_i - 4B_i\mathcal{A}_i\eta_i\xi_i + 4a_iB_i\mathcal{A}_i\eta_i\xi_i + 4B_i^2\mathcal{A}_i\eta_i\xi_i - 4B_ix_iy_i\mathcal{A}_i^2\eta_i\xi_i + 4B_i\mathcal{A}_i\beta_i\eta_i\xi_i - 4B_i^2\mathcal{A}_i\beta_i\eta_i\xi_i + 6y_i\mathcal{A}_i^2\eta_i^2\xi_i - 2B_iy_i\mathcal{A}_i^2\eta_i^2\xi_i - 2B_i^2x_i^2\mathcal{A}_i^2\xi_i^2 + 6B_ix_i\mathcal{A}_i^2\eta_i\xi_i^2 - 2B_i^2x_i\mathcal{A}_i^2\eta_i\xi_i^2 - 3\mathcal{A}_i^2\eta_i^2\xi_i^2 + 4B_i\mathcal{A}_i^2\eta_i^2\xi_i^2 - B_i^2\mathcal{A}_i^2\eta_i^2\xi_i^2\big)\epsilon + O[\epsilon]^2\Big]$$

```
Block[{i, j, k},
 dΔ_{i_→j_,k_} = (bΔ_{i→3,1} aΔ_{i→2,4}) ~B_{1,2,3,4}~ (dm_{3,4→k} dm_{1,2→j})]
```

$$\mathbb{E}\Big[a_j\alpha_i + a_k\alpha_i + b_j\beta_i + b_k\beta_i,\ y_j\eta_i + B_jy_k\eta_i + x_j\xi_i + x_k\xi_i,\ 1 + \frac{1}{2}\big(B_jy_jy_k\eta_i^2 - 2a_jx_k\xi_i + x_jx_k\xi_i^2\big)\epsilon + O[\epsilon]^2\Big]$$

```
R_{i_,j_} := 𝔼[b_i a_j, y_i x_j, 1 - ϵ y_i^2 x_j^2 /4 + O[ϵ]^2]
Block[{i, j}, R̄_{i_,j_} = Expand /@ R_{i,j}~B_j~dS_j]
```

$$\mathbb{E}\Big[-a_jb_i,\ -\frac{x_jy_i}{B_i},\ 1 + \frac{(-4a_jB_ix_jy_i - 3x_j^2y_i^2)\epsilon}{4B_i^2} + O[\epsilon]^2\Big]$$

```
Block[{i}, {
  u_{i_} = R_{1,2}~B_1~dS_1~B_{1,2}~dm_{2,1→i},
  ui_{i_} := R_{1,2}~B_2~dS_2~B_2~dS_2~B_{1,2}~dm_{2,1→i}
 }]
```

$$\Big\{\mathbb{E}\Big[-a_ib_i,\ -\frac{x_iy_i}{B_i},\ B_i + \frac{(-4a_iB_i^2 - 4B_ix_iy_i - 4a_iB_ix_iy_i - 3x_i^2y_i^2)\epsilon}{4B_i} + O[\epsilon]^2\Big],\ \text{Null}\Big\}$$

```
Block[{i},
 {CC_{i_} = 𝔼[0, 0, B_i^{1/2} e^{-ϵ a_i/2} + O[ϵ]^2],
  C̄C̄_{i_} = 𝔼[0, 0, B_i^{-1/2} e^{ϵ a_i/2} + O[ϵ]^2]
 }]
```

$$\Big\{\mathbb{E}\Big[0,0,\sqrt{B_i} - \frac{1}{2}\big(a_i\sqrt{B_i}\big)\epsilon + O[\epsilon]^2\Big],\ \mathbb{E}\Big[0,0,\frac{1}{\sqrt{B_i}} + \frac{a_i\epsilon}{2\sqrt{B_i}} + O[\epsilon]^2\Big]\Big\}$$

```
Block[{i, j}, {
  Kink_{i_} = (R_{1,3} C̄C̄_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→i},
  K̄ink_{j_} = (R̄_{1,3} CC_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→j}
 }]
```

$$\Big\{\mathbb{E}\Big[a_ib_i,\ x_iy_i,\ \frac{1}{\sqrt{B_i}} + \frac{(2a_i - x_i^2y_i^2)\epsilon}{4\sqrt{B_i}} + O[\epsilon]^2\Big],\ \mathbb{E}\Big[-a_jb_j,\ -\frac{x_jy_j}{B_j},\ \sqrt{B_j} + \frac{(-2a_jB_j^2 - 4a_jB_jx_jy_j - 3x_j^2y_j^2)\epsilon}{4B_j^{3/2}} + O[\epsilon]^2\Big]\Big\}$$

```
Z = R_{1,5} R_{6,2} R_{3,7} C̄C̄_4 K̄ink_8 K̄ink_9 K̄ink_10;
Do[Z = Z~B_{1,r}~dm_{1,r→1}, {r, 2, 10}];
Simplify /@ Z
```

$$\mathbb{E}\Big[0,0,\ \frac{B_1}{1 - B_1 + B_1^2} + \frac{1}{(1 - B_1 + B_1^2)^3}B_1\big(-B_1 + 2B_1^2 + 2B_1^4 + a_1(-1 + B_1 - B_1^3 + B_1^4) - 2x_1y_1 - B_1^3(3 + 2x_1y_1)\big)\epsilon + O[\epsilon]^2\Big]$$

```
b2t_{i_} := 𝔼[α_i a_i - β_i t_i, ξ_i x_i + η_i y_i, 1 + ϵ β_i a_i + O[ϵ]^2]
t2b_{i_} := 𝔼[α_i a_i - τ_i b_i, ξ_i x_i + η_i y_i, 1 + ϵ τ_i a_i + O[ϵ]^2]
```

Faddeev's Formula (In as much as we can tell, first appeared without proof in Faddeev [Fa], rediscovered and proven in Quesne [Qu], and again with easier proof, in Zagier [Za]). With $[n]_q := \frac{q^n-1}{q-1}$, with $[n]_q! := [1]_q[2]_q\cdots[n]_q$ and with $\mathbb{e}_q^x := \sum_{n\geq 0}\frac{x^n}{[n]_q!}$, we have

$$\log \mathbb{e}_q^x = \sum_{k\geq 1}\frac{(1-q)^kx^k}{k(1-q^k)} = x + \frac{(1-q)^2x^2}{2(1-q^2)} + \ldots.$$

Proof. We have that $\mathbb{e}_q^x = \frac{\mathbb{e}_q^{qx}-\mathbb{e}_q^x}{qx-x}$ ("the q-derivative of $\mathbb{e}_q^x$ is itself"), and hence $\mathbb{e}_q^{qx} = (1+(1-q)x)\mathbb{e}_q^x$, and

$$\log\mathbb{e}_q^{qx} = \log(1+(1-q)x) + \log\mathbb{e}_q^x.$$

Writing $\log\mathbb{e}_q^x = \sum_{k\geq 1}a_kx^k$ and comparing powers of x, we get $q^ka_k = -(1-q)^k/k + a_k$, or $a_k = \frac{(1-q)^k}{k(1-q^k)}$. □

A Partial To Do List.

- Complete all "docility" arguments by identifying a "contained" docile substructure.
- Understand denominators and get rid of them.
- See if much can be gained by including P in the exponential: $\mathbb{e}^{L+Q}P \rightsquigarrow \mathbb{e}^{L+Q+P}$?
- Clean the program and make it efficient.
- Run it for all small knots and links, at $k = 2, 3$.
- Understand the centre and figure out how to read the output.
- Extend to sl_3 and beyond.
- Prove a genus bound and a Seifert formula.
- Obtain "Gauss-Gassner formulas" (ωεβ/NCSU).
- Relate with Melvin-Morton-Rozansky and with Rozansky-Overbay.
- Understand the braid group representations that arise.
- Find a topological interpretation. The Garoufalidis-Rozansky "loop expansion" [GR]?
- Figure out the action of the Cartan automorphism.
- Disprove the ribbon-slice conjecture!
- Figure out the action of the Weyl group.
- Do everything at the "arrow diagram" level of finite-type invariants of (rotational) virtual tangles.
- What else can you do with the "solvable approximations"?
- And with the "Gaussian zip and bind" technology?

Further References.

[GR] S. Garoufalidis and L. Rozansky, *The Loop Exapnsion of the Kontsevich Integral, the Null-Move, and S-Equivalence*, arXiv:math.GT/0003187.

[Fa] L. Faddeev, *Modular Double of a Quantum Group*, arXiv:math/9912078.

[Qu] C. Quesne, *Jackson's q-Exponential as the Exponential of a Series*, arXiv:math-ph/0305003.

[Za] D. Zagier, *The Dilogarithm Function*, in Cartier, Moussa, Julia, and Vanhove (eds) *Frontiers in Number Theory, Physics, and Geometry II*. Springer, Berlin, Heidelberg, and ωεβ/Za.